<commit_before>{ lib
, stdenv
, rustPlatform
, fetchFromGitHub
, help2man
, installShellFiles
, libiconv
, Security
, CoreServices
, nix-update-script
}:

rustPlatform.buildRustPackage rec {
  pname = "texlab";
  version = "4.2.1";

  src = fetchFromGitHub {
    owner = "latex-lsp";
    repo = pname;
    rev = "refs/tags/v${version}";
    sha256 = "sha256-za3TauhNoxGDphpY615EnTt46HpMgS+sYpBln/twefw=";
  };

  cargoSha256 = "sha256-wppaa3IGOqkFu/1CAp8g+PlPtMWm/7qNpPu0k4/mL3A=";

  outputs = [ "out" "man" ];

  nativeBuildInputs = [
    help2man
    installShellFiles
  ];

  buildInputs = lib.optionals stdenv.isDarwin [
    libiconv
    Security
    CoreServices
  ];

  postInstall = ''
    # TexLab builds man page separately in CI:
    # https://github.com/latex-lsp/texlab/blob/v4.2.1/.github/workflows/publish.yml#L131-L135
    help2man --no-info "$out/bin/texlab" > texlab.1
    installManPage texlab.1

    # Remove generated dylib of human_name dependency. TexLab statically
    # links to the generated rlib and doesn't reference the dylib. I
    # couldn't find any way to prevent building this by passing cargo flags.
    # See https://github.com/djudd/human-name/blob/master/Cargo.toml#L43
    rm "$out/lib/libhuman_name${stdenv.hostPlatform.extensions.sharedLibrary}"
    rmdir "$out/lib"
  '';

  passthru.updateScript = nix-update-script {
    attrPath = pname;
  };

  meta = with lib; {<commit_msg>texlab: fix cross-compilation<commit_after>{ lib
, stdenv
, rustPlatform
, fetchFromGitHub
, help2man
, installShellFiles
, libiconv
, Security
, CoreServices
, nix-update-script
}:

let
  isCross = stdenv.hostPlatform != stdenv.buildPlatform;
in rustPlatform.buildRustPackage rec {
  pname = "texlab";
  version = "4.2.1";

  src = fetchFromGitHub {
    owner = "latex-lsp";
    repo = pname;
    rev = "refs/tags/v${version}";
    sha256 = "sha256-za3TauhNoxGDphpY615EnTt46HpMgS+sYpBln/twefw=";
  };

  cargoSha256 = "sha256-wppaa3IGOqkFu/1CAp8g+PlPtMWm/7qNpPu0k4/mL3A=";

  outputs = [ "out" ] ++ lib.optional (!isCross) "man";

  nativeBuildInputs = [ installShellFiles ]
  ++ lib.optional (!isCross) help2man;

  buildInputs = lib.optionals stdenv.isDarwin [
    libiconv
    Security
    CoreServices
  ];

  postInstall = ''
    # Remove generated dylib of human_name dependency. TexLab statically
    # links to the generated rlib and doesn't reference the dylib. I
    # couldn't find any way to prevent building this by passing cargo flags.
    # See https://github.com/djudd/human-name/blob/master/Cargo.toml#L43
    rm "$out/lib/libhuman_name${stdenv.hostPlatform.extensions.sharedLibrary}"
    rmdir "$out/lib"
  ''
  # When we cross compile we cannot run the output executable to
  # generate the man page
  + lib.optionalString (!isCross) ''
    # TexLab builds man page separately in CI:
    # https://github.com/latex-lsp/texlab/blob/v4.2.1/.github/workflows/publish.yml#L131-L135
    help2man --no-info "$out/bin/texlab" > texlab.1
    installManPage texlab.1
  '';

  passthru.updateScript = nix-update-script {
    attrPath = pname;
  };

  meta = with lib; {
{'fraction_non_alphanumeric': 0.12032258064516128, 'fraction_numerical': 0.02838709677419355, 'mean_word_length': 3.4490674318507892, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 5, 'https://': 4, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '362467', 'n_tokens_mistral': 1219, 'n_tokens_neox': 1144, 'n_words': 229}